Laplace transform

with *Mathematica*

Dr. Luigi E. Masciovecchio
luigi.masciovecchio@virgilio.it
available as PDF on http://sites.google.com/site/luigimasciovecchio/

```
Print["Revision ", IntegerPart[Date[]]]
```

```
Revision {2013, 8, 29, 7, 12, 31}
```

```
Print["This system is:"]
{"ProductIDName", "ProductVersion"} /. $ProductInformation
ReadList["!ver", String][[2]]
{$MachineType, $ProcessorType, $ByteOrdering, $SystemCharacterEncoding}
```

```
This system is:

{Mathematica, 5.2 for Microsoft Windows (June 20, 2005)}

Windows 98 [versione 4.10.1998]

{PC, x86, -1, WindowsANSI}
```

5.2.2 La trasformata di Laplace

■ The Laplace transform of a function $f(t)$ is defined to be $\int_{0-}^{\infty} f(t)\, e^{-s\,t}\, dt$.

IMPORTANT: The lower limit of the integral is effectively taken to be 0-, so that the Laplace transform of the Dirac delta function δ(t) is equal to 1.

■ The inverse Laplace transform of a function $F(s)$ is defined to be $\frac{1}{2\,\pi\,i} \int_{\gamma-i\,\infty}^{\gamma+i\,\infty} F(s)\, e^{s\,t}\, ds$,

where γ is an arbitrary positive constant chosen so that the contour of integration lies to the right of all singularities in F(s).

The Laplace transform is an integral transform perhaps second only to the Fourier transform in its utility in solving physical problems. Laplace transforms have the property that they turn integration and differentiation into essentially algebraic operations. This property can be used to transform differential equations into algebraic equations, which can then be inverse transformed to obtain the solution.

$$\left\{\int_{-\infty}^{\infty} \mathbf{DiracDelta}[x]\, dx,\ \int_{0-\$MachineEpsilon}^{\infty} \mathbf{DiracDelta}[x]\, dx,\ \int_{0}^{\infty} \mathbf{DiracDelta}[x]\, dx,\ \int_{1}^{\infty} \mathbf{DiracDelta}[x]\, dx\right\}$$

$$\left\{1,\ 1.,\ \frac{1}{2},\ 0\right\}$$

- ***Trasformata di Laplace di funzioni notevoli***

(cfr. De Santis, p. 506, tab. 6.2.1)

```
fns = {1, a, t, t^2/2, Exp[-a t], t Exp[-a t], 1 - Exp[-a t], Cos[ω t], UnitStep[t]};
LaplaceTransform[fns, t, s] // Factor // TraditionalForm
```

$$\left\{\frac{1}{s}, \frac{a}{s}, \frac{1}{s^2}, \frac{1}{s^3}, \frac{1}{a+s}, \frac{1}{(a+s)^2}, \frac{a}{s(a+s)}, \frac{s}{s^2+\omega^2}, \frac{1}{s}\right\}$$

```
LaplaceTransform[f'[t], t, s] // TraditionalForm
```

$$s(\mathcal{L}_t[f(t)](s)) - f(0)$$

```
LaplaceTransform[∫_{-ϵ}^t f[τ] dτ, t, s] // Distribute // TraditionalForm
```

$$\frac{\mathcal{L}_t[f(t)](s)}{s} - \frac{\text{Integrate}[f(t), \{t, 0, -\epsilon\}, \text{Assumptions} \to s > 0]}{s}$$

- ***Un primo esempio semplice***

```
f[t_] = t^2 Sin[t];
F[s_] = LaplaceTransform[f[t], t, s];
{f[t], F[s], F[a + b ⅈ], F[-.5 + 0.5 ⅈ]}
```

$$\left\{t^2 \text{Sin}[t], \frac{-2+6 s^2}{(1+s^2)^3}, \frac{-2+6(a+ⅈ b)^2}{(1+(a+ⅈ b)^2)^3}, 1.856 - 1.792\, ⅈ\right\}$$

```
PR = {Automatic, Automatic, {-2, 2}};
pf = Plot[f[t], {t, 0, 40}, AxesLabel → {"t", "f(t)"}, DisplayFunction → Identity];
pReF = DensityPlot[Re[F[a + b ⅈ]], {a, -2, 2}, {b, -2, 2},
  PlotPoints → 100, Mesh → False, PlotRange → PR, FrameLabel → {"Re s", "Im s"},
  PlotLabel → "Re F(s)", DisplayFunction → Identity];
pImF = DensityPlot[Im[F[a + b ⅈ]], {a, -2, 2}, {b, -2, 2}, PlotPoints → 100,
   Mesh → False, PlotRange → PR, FrameLabel → {"Re s", "Im s"},
   PlotLabel → "Im F(s)", DisplayFunction → Identity];
pCode = DensityPlot[y, {x, 0, 1}, {y, -2, 2}, PlotPoints → 100, Mesh → False,
  FrameTicks → {None, Automatic}, AspectRatio → Automatic, DisplayFunction → Identity];
Show[GraphicsArray[{pf, pReF, pImF, pCode}, GraphicsSpacing → -0.15]];
```



```
p1 = Plot3D[Re[F[a + b ⅈ]], {a, -2, 2}, {b, -2, 2}, PlotPoints → 100, Mesh → False,
    PlotRange → PR, AxesLabel → {"Re s", "Im s", "Re F(s)"}, DisplayFunction → Identity];
p2 = Plot3D[Im[F[a + b ⅈ]], {a, -2, 2}, {b, -2, 2}, PlotPoints → 100, Mesh → False,
    PlotRange → PR, AxesLabel → {"Re s", "Im s", "Im F(s)"}, DisplayFunction → Identity];
Show[GraphicsArray[{p1, p2}]];
```



- ***Un secondo esempio meno semplice***

```
f[t_] = 1 / (1 + t^2);
F[s_] = LaplaceTransform[f[t], t, s];
{f[t], F[s], F[a + b ⅈ], F[0.5 + 1.5 ⅈ]} // TraditionalForm
```

$$\left\{\frac{1}{t^2+1},\ \mathrm{Ci}(s)\sin(s)+\frac{1}{2}\cos(s)\,(\pi-2\,\mathrm{Si}(s)),\ \mathrm{Ci}(a+i\,b)\sin(a+i\,b)+\frac{1}{2}\cos(a+i\,b)\,(\pi-2\,\mathrm{Si}(a+i\,b)),\ 0.385562-0.432023\,i\right\}$$

$$\textbf{Definitions:}\ \mathrm{Ci}(z)=-\int_z^{\infty}\cos(t)/t\,dt,\quad \mathrm{Si}(z)=\int_0^z\sin(t)/t\,dt.$$

```
PR = {Automatic, Automatic, {-2, 2}};
pf = Plot[f[t], {t, 0, 4}, AxesLabel → {"t", "f(t)"}, DisplayFunction → Identity];
pReF = DensityPlot[Re[F[a + b ⅈ]], {a, -2, 2}, {b, -2, 2},
   PlotPoints → 20, Mesh → False, PlotRange → PR, FrameLabel → {"Re s", "Im s"},
   PlotLabel → "Re F(s)", DisplayFunction → Identity];
pImF = DensityPlot[Im[F[a + b ⅈ]], {a, -2, 2}, {b, -2, 2}, PlotPoints → 20,
    Mesh → False, PlotRange → PR, FrameLabel → {"Re s", "Im s"},
    PlotLabel → "Im F(s)", DisplayFunction → Identity];
pCode = DensityPlot[y, {x, 0, 1}, {y, -2, 2}, PlotPoints → 20, Mesh → False,
   FrameTicks → {None, Automatic}, AspectRatio → Automatic, DisplayFunction → Identity];
Show[GraphicsArray[{pf, pReF, pImF, pCode}, GraphicsSpacing → -0.15]];
```



- ***Il motore elettrico in corrente continua (soluzione di equazioni integro-differenziali)***

(De Santis, p. 95-103 e p. 470-478)

Nota : coppia di disturbo $C_d[t] \equiv C_{costante} + C_{random}[t]$, momento d'inerzia totale $J_m + J_c = J$

```
ModelloMotoreCC =
 {          va[t] == Ra ia[t] + La ia'[t] + Ke ω[t],
  (Jm + Jc) ω'[t] == KT ia[t] - Kav ω[t] - Cd[t]    }
```

$\{v_a[t] == K_e\,\omega[t] + R_a\,i_a[t] + L_a\,i_a'[t],\ (J_c + J_m)\,\omega'[t] == -K_{av}\,\omega[t] - C_d[t] + K_T\,i_a[t]\}$

(cfr. De Santis, p. 97, eq. 1.12.6-8)

```
DSolve[{(Jm + Jc) ω'[t] == -Kav ω[t], ω[0] == ϖ0}, ω[t], t] // FullSimplify
```

$\left\{\left\{\omega[t] \to e^{-\frac{t K_{av}}{J_c + J_m}}\,\varpi_0\right\}\right\}$

```
LModelloMotoreCC = LaplaceTransform[ModelloMotoreCC, t, s] /.
  {LaplaceTransform[Cd[t], t, s] → Cd[s], LaplaceTransform[ω[t], t, s] → ω[s],
   LaplaceTransform[ia[t], t, s] → ia[s], LaplaceTransform[va[t], t, s] → va[s]}
```

$\{v_a[s] == K_e\,\omega[s] + R_a\,i_a[s] + L_a\,(-i_a[0] + s\,i_a[s]),$
$(J_c + J_m)\,(-\omega[0] + s\,\omega[s]) == -K_{av}\,\omega[s] - C_d[s] + K_T\,i_a[s]\}$

(cfr. De Santis p. 98, eq. 1.12.9-11)

```
Lω[s_] = Collect[Solve[LModelloMotoreCC, ω[s], ia[s]][[1, 1, 2]], {Cd[s], va[s]}];
Lia[s_] = Collect[Solve[LModelloMotoreCC, ia[s], ω[s]][[1, 1, 2]], {Cd[s], va[s]}];
HoldForm[ω[s]] == Lω[s]
Print["(cfr. De Santis, p. 100, eq. 1.12.13)"]
HoldForm[ia[s]] == Lia[s]
```

$$\omega[s] == -\frac{(s L_a + R_a)\,C_d[s]}{K_e K_T + s^2 J_c L_a + s^2 J_m L_a + s K_{av} L_a + s J_c R_a + s J_m R_a + K_{av} R_a} - \frac{-s J_c L_a\,\omega[0] - s J_m L_a\,\omega[0] - J_c R_a\,\omega[0] - J_m R_a\,\omega[0] - K_T L_a\,i_a[0]}{K_e K_T + s^2 J_c L_a + s^2 J_m L_a + s K_{av} L_a + s J_c R_a + s J_m R_a + K_{av} R_a} + \frac{K_T\,v_a[s]}{K_e K_T + s^2 J_c L_a + s^2 J_m L_a + s K_{av} L_a + s J_c R_a + s J_m R_a + K_{av} R_a}$$

(cfr. De Santis, p. 100, eq. 1.12.13)

$$i_a[s] == \frac{K_e\,C_d[s]}{K_e K_T + s^2 J_c L_a + s^2 J_m L_a + s K_{av} L_a + s J_c R_a + s J_m R_a + K_{av} R_a} - \frac{J_c K_e\,\omega[0] + J_m K_e\,\omega[0] - s J_c L_a\,i_a[0] - s J_m L_a\,i_a[0] - K_{av} L_a\,i_a[0]}{K_e K_T + s^2 J_c L_a + s^2 J_m L_a + s K_{av} L_a + s J_c R_a + s J_m R_a + K_{av} R_a} - \frac{(-s J_c - s J_m - K_{av})\,v_a[s]}{K_e K_T + s^2 J_c L_a + s^2 J_m L_a + s K_{av} L_a + s J_c R_a + s J_m R_a + K_{av} R_a}$$

Caratteristiche tecniche del motore in corrente continua tipo M9234 (Deltaomega), v. De Santis, p. 102, tab. 1.12.1)

```
ParametriDeltaOmegaM9234 = {Va := 12.0 (*V*), Ias := 14.5 (*A*), Ra := Va / Ias
   (*=0.828 Ω*), La := 0.6 10^-3 (*H*), Jm := 41.7 10^-7 (*kg m^2*), Kav := 2.61 10^-6
   (*m s N*), KT := 18.3 10^-3 (*m N A^-1*), Ke := 18.1 10^-3 (*V s rad^-1*)};
Print["costante di tempo elettrica τe = ", (Ra / La)^-1 10^3, " ms, ",
 "costante di tempo meccanica τm = ", (KT Ke / (Ra Jm))^-1 10^3, " ms"]
```

costante di tempo elettrica τ_e = 0.725 ms, costante di tempo meccanica τ_m = 10.4188 ms

```
vaSim[t_] = Va UnitStep[t]; va[s_] = LaplaceTransform[vaSim[t], t, s];
CdSim[t_] = 0.; Cd[s_] = LaplaceTransform[CdSim[t], t, s];
ParametriSim =
  {Jc := 0.0 (*kg m^2*), ω[0] := 0.0 (*rad s^-1*), ia[0] := 0.0 (*A*)};
Lω[s] // Factor
ω1[t_] = InverseLaplaceTransform[Lω[s], s, t]
Lia[s] // Factor
InverseLaplaceTransform[Lia[s], s, t]
La = 50 La
ω2[t_] = InverseLaplaceTransform[Lω[s], s, t]
```

$$\frac{8.77698\times 10^{7}}{s\,(104.471+s)\,(1275.47+s)}$$

$$658.688 + 58.7654\, e^{-1275.47\, t} - 717.453\, e^{-104.471\, t}$$

$$\frac{20000.\,(0.625899+1.\,s)}{s\,(104.471+s)\,(1275.47+s)}$$

$$0.093944 - 17.0711\, e^{-1275.47\, t} + 16.9772\, e^{-104.471\, t}$$

$$0.03$$

$$658.688 - (329.344 + 93.5531\, i)\, e^{(-14.1061-49.6589\, i)\, t} - (329.344 - 93.5531\, i)\, e^{(-14.1061+49.6589\, i)\, t}$$

```
Show[GraphicsArray[{Plot[ω1[t], {t, 0, 0.05},
      PlotRange → {0, 1000}, AxesLabel → {"t (s)", "ω(t) (rad/s)"},
      PlotLabel -> "La = 0.6 10^-3 H", DisplayFunction → Identity],
    Plot[ω2[t], {t, 0, 0.5}, PlotRange → {0, 1000}, AxesLabel → {"t (s)", "ω(t) (rad/s)"},
      PlotLabel -> "La = 3.0 10^-2 H", PlotLabel -> "La  H", DisplayFunction → Identity]}]];
```



When someone says, "I want a programming language in which I need only say what I wish done", give him a lollipop.
(Alan Perlis)